
// "An enhanced easier inverted pendulum balancing robot"

// You need only an hour to enhance it, if you have three Materials.

// (This sketch is ver.2.1.d for a digital output gyroscope.)

// No timer library is used in this version.

// But stability of robot is more improved than earlier version.

// A sample receiving sketch of Processing should be rewrite in a few lines. 

// Copyright (C) 2014 ArduinoDeXXX All Rights Reserved. 

byte countS = 0;

int recOmegaI[10];

int omegaI = 0;

long thetaI = 0;

long sumPower = 0; 

long sumSumP = 0; 

const int kAngle = 65;

const int kOmega = 360;

const long kSpeed = 65;

const long kDistance = 350;

long powerScale;

int power;

long vE5 = 0;

long xE5 = 0;

/*** added (1) ********************/

int c1 = 0;

int c2 = 0;

int ax, ay, az;

long R;

/*** added (1): end ***************/

#include <SPI.h>

void L3GD20_write(byte reg, byte val) {

  digitalWrite(10, LOW);

  SPI.transfer(reg);

  SPI.transfer(val);

  digitalWrite(10, HIGH);

}

byte L3GD20_read(byte reg) {

  byte ret = 0;

  digitalWrite(10, LOW);

  SPI.transfer(reg | 0x80);

  ret = SPI.transfer(0);

  digitalWrite(10, HIGH);

  return ret;

}

/*** added (2) ********************/

void LIS3DH_write(byte reg, byte val) {

  digitalWrite(3, LOW);

  SPI.transfer(reg & 0b00111111);

  SPI.transfer(val);

  digitalWrite(3, HIGH);

}

byte LIS3DH_read(byte reg) {

  byte ret = 0;

  digitalWrite(3, LOW);

  SPI.transfer(reg | 0b10000000);

  ret = SPI.transfer(0);

  digitalWrite(3, HIGH);

  return ret;

}

/*** added (2): end ***************/

void setup () {



  Serial .begin(115200);

  pinMode(4, OUTPUT);

  pinMode(5, OUTPUT);

  pinMode(6, OUTPUT);

  pinMode(7, OUTPUT);

  pinMode(8, OUTPUT);

  pinMode(9, OUTPUT);

  for ( int i = 0 ; i < 10 ; i++ ) { recOmegaI[i] = 0; }

/*** added (3) ********************/

  pinMode(3, OUTPUT);

  digitalWrite(3, HIGH);

/*** added (3): end ***************/

  pinMode(10, OUTPUT);

  digitalWrite(10, HIGH);

  SPI.begin();

  SPI.setBitOrder(MSBFIRST);

  SPI.setDataMode(SPI_MODE3);

  SPI.setClockDivider(SPI_CLOCK_DIV2);

  L3GD20_write(0x20, B11001111);

  L3GD20_write(0x23, B00000000);

/*** added (4) ********************/

  LIS3DH_write(0x20, B10010111);

  LIS3DH_write(0x23, B00001000);

/*** added (4): end ***************/

  delay(300);

}

void loop () {

  chkAndCtl();

/*** added (5) ********************/

  R=0;

  c1 = 0;

  c2 = 0;

/*** added (5): end ***************/

  if ( power > 0 ) {  

    analogWrite( 6, power );

    digitalWrite( 4, HIGH );        

    digitalWrite( 5, LOW );  

    analogWrite( 9, power );

    digitalWrite( 7, HIGH );        

    digitalWrite( 8, LOW );  

/*** added (6) ********************/

    for ( int i = 0 ; i < 20 ; i++ ) {

      c1 = c1 + analogRead(A1);

      c2 = c2 + analogRead(A2);

      R = R + ( (L3GD20_read(0x2B) << 8) | L3GD20_read(0x2A) );

    }

/*** added (6): end ***************/

  } else {

    analogWrite( 6, - power );

    digitalWrite( 4, LOW );        

    digitalWrite( 5, HIGH );  

    analogWrite( 9, - power );

    digitalWrite( 7, LOW );        

    digitalWrite( 8, HIGH );  

/*** added (7) ********************/



    for ( int i = 0 ; i < 20 ; i++ ) {

      c1 = c1 - analogRead(A1);

      c2 = c2 - analogRead(A2);

      R = R + ( (L3GD20_read(0x2B) << 8) | L3GD20_read(0x2A) );

    }

/*** added (7): end ***************/

  }

/*** added (8) ********************/

  omegaI = R * 0.00875 / 20; //The similar line to this line is found in "chkAndCtl()" in the original sketch.

  Serial .print(millis());  Serial .print(",");

  Serial .print(kAngle * thetaI / 100);Serial .print(",");

  Serial .print(kOmega * omegaI / 100); Serial .print(",");

  Serial .print(kSpeed * vE5 / 1000); Serial .print(",");

  Serial .print(kDistance * xE5 / 1000);Serial .print(",");

  Serial .print(c1/20);  Serial .print(",");

  Serial .print(c2/20);  Serial .print(",");

  Serial .print(ax);  Serial .print(",");

  Serial .print(ay);  Serial .print(",");

  Serial .println(az); 

/*** added (8): end ***************/

}

void chkAndCtl() {

/*** added (9) ********************/

  ax = ( (LIS3DH_read(0x29) << 8) | LIS3DH_read(0x28) );

  ay = ( (LIS3DH_read(0x2B) << 8) | LIS3DH_read(0x2A) );

  az = ( (LIS3DH_read(0x2D) << 8) | LIS3DH_read(0x2C) );

/*** added (9): end ***************/

  if ( abs( omegaI ) < 2 ) { omegaI = 0; }

  recOmegaI[0] = omegaI;

  thetaI = thetaI + omegaI;

  countS = 0;  

  for ( int i = 0 ; i < 10 ; i++ ) {

    if ( abs( recOmegaI[i] ) < 4 ) { countS++; }

  }

  if ( countS > 9 ) {

    thetaI = 0;

    vE5 = 0;

    xE5 = 0;

    sumPower = 0;

    sumSumP = 0;

  }

  for ( int i = 9 ; i > 0 ; i-- ) { recOmegaI[ i ] = recOmegaI[ i-1 ]; }    

  powerScale = ( kAngle * thetaI / 100 ) + ( kOmega * omegaI / 100 ) + ( kSpeed * vE5 / 1000 ) + ( kDistance * xE5 / 1000 );

  power = max ( min ( powerScale , 255 ) , -255 );  //This line is different from the original sketches.

  sumPower = sumPower + ( ( c1 + c2 ) / 20 / 0.552);  //This line is different from the original sketches.

  sumSumP = sumSumP + sumPower;

  vE5 = sumPower;

  xE5 = sumSumP / 1000;

}

